Aberger M., Otter M.:

Modeling Friction in Modelica with the Lund-Grenoble Friction Model

2nd International Modelica Conference, Proceedings, pp. 285-294

Paper presented at the 2nd International Modelica Conference, March 18-19, 2002,
Deutsches Zentrum für Luft- und Raumfahrt e.V. (DLR), Oberpfaffenhofen, Germany.

Modeling Friction in Modelica with the Lund-Grenoble Friction Model

Martin Aberger
Johannes Kepler University Linz,
Department for Design and Control of
Mechatronical Systems
A-4040 Linz, Austria
martin.aberger@students.jku.at

Martin Otter
DLR Oberpfaffenhofen,
Institute of Robotics and Mechatronics,
D-82230 Wessling, Germany
Martin.Otter@dlr.de

Abstract

The properties of the Lund-Grenoble friction model are summarized and different types of friction elements - bearing friction, clutch, one-way clutch, are implemented in Modelica using this friction formulation. The dynamic properties of these components are determined in simulations and compared with the friction models available in the Modelica standard library. This includes also an automatic gearbox model where 6 friction elements are coupled dynamically.

1 The rotational LuGre Model

1.1 Model Derivation

The LuGre (Lund-Grenoble) model [2] is a dynamic friction model with the relative angular velocity between the two surfaces in contact ω as input, and the friction torque τ as output. It approximates friction as a phenomenon caused by bristles in contact. The model can be seen as an extension of the simplified Dahl model. The LuGre model is described in standard form by a first-order nonlinear differential equation, see [6],

$$\frac{dz}{dt} = \omega - \frac{|\omega|}{g(\omega)} z \tag{1}$$

$$g(\omega) = \frac{1}{\sigma_0}\left(\tau_C + (\tau_S - \tau_C) e^{-(\omega/\omega_s)^2} \right) \tag{2}$$

$$\tau = \sigma_0 z + \sigma_1(\omega)\frac{dz}{dt} + \tau_v \omega \tag{3}$$

$$\sigma_1(\omega) = \sigma_1 e^{-(\omega/\omega_d)^2} . \tag{4}$$

Where z denotes the average bristle deflection, σ_0 is the stiffness of the bristles, $\sigma_1(\omega)$ is the damping coefficient, ω_d describes the velocity interval around zero for which the damping is active, τ_v is the linear viscous friction coefficient, τ_C is the Coulomb friction level, τ_S is the static friction level, $(\tau_S \geq \tau_C)$, and ω_S is the Stribeck velocity. The function $g(\omega)$ defines how the average deflection depends on the relative velocity between the two surfaces. The simplified form of the LuGre model is given by a constant damping coefficient

$$\sigma_1(\omega) = \sigma_1 . \tag{5}$$

With a constant damping coefficient and $(\tau_S > \tau_C)$ the model is dissipativ, see [1], if and only if

$$\sigma_1 \leq \frac{\tau_C \tau_v}{\tau_S - \tau_C} . \tag{6}$$

The velocity dependent damping coefficient $\sigma_1(\omega)$ was not implemented in the Modelica model because no identified or measured parameters for ω_d have been found in the literature. It is already rather difficult to identify the dynamic parameters σ_0 and σ_1, which is also stated in [5]. With an increasing number of parameters the complexity of the identification process rises.

For steady-state motion $dz/dt = 0$ the average bristle deflection is given by, see (1)

$$z_{ss} = \text{sgn}(\omega) g(\omega) \tag{7}$$

Hence the relation between angular velocity and friction torque for steady state motion is

$$\begin{aligned} \tau_{ss} &= \sigma_0 \,\text{sgn}(\omega) g(\omega) + \tau_v \omega \\ &= \left(\tau_C + (\tau_S - \tau_C) e^{-(\omega/\omega_s)^2} \right) \text{sgn}(\omega) + \tau_v \omega \end{aligned} . \tag{8}$$

If the angular velocity is not zero when integration starts, the initial value of z, see (1), shall be computed such that $dz/dt = 0$ for $t = 0$ in order to avoid (non-physical) peaks in the friction torque at the start of the simulation. Simulation experiments with stick-slip motion show that integration methods with variable step-size may have difficulties to compute the break away torque in certain cases, especially if the relative tolerance is not set strictly enough. To improve the reliability and accuracy of the simulation, an auxiliary Boolean equation is introduced

$$zs_{neg} = \begin{cases} \text{true} & \text{if } \dfrac{dz}{dt} < 0 \\ \text{false} & \text{otherwise} \end{cases} . \tag{9}$$

that triggers a state event when dz/dt changes sign (in Modelica, every value change of a relation triggers an event). Additionally, it is useful to scale the average bristle deflection *z,* as it is small compared to other state variables. Both effects are discussed in more detail below. A reasonable choice for the scaling parameter z_N is

$$z_N \approx \frac{1}{\sigma_0}. \tag{10}$$

The parameters of the LuGre friction model depend mostly on the application, especially the friction torque levels τ_C, τ_S, and τ_v may vary widely. Reasonable choices for the other parameters are given in [6]

$$\omega_S = 0.01$$
$$\sigma_0 = 10^3 - 10^5$$
$$\sigma_1 = 2\sqrt{\sigma_0 J}$$

where J is the inertia of the body subject to friction. The time for integration depends heavily on the choice of the dynamic parameters σ_0 and σ_1 because they determine essentially the stiffness of the differential equation.

1.2 Dynamic Model Behaviour

To verify the dynamic behaviour of the simplified LuGre model, the same simulations as in [2] are performed using Dymola as simulation engine [4] and a Modelica coded LuGre model. The *final* results agree qualitatively with the results in [2]. Stick-slip motion is a typical behaviour of systems with friction. It is caused by the fact that friction is larger at rest than during start of sliding.

The experiment is shown in Figure 1. An inertia with J=1 kgm^2 with friction to the ground is connected to a spring with stiffness k=2 Nm/rad. The end of the spring is rotating with a small constant velocity of ω=0.1 rad/s. The inertia is originally at rest and the torque from the spring increases linearly. The friction torque counteracts the spring torque, and a small displacement follows. When the applied torque reaches the break away torque, in this case approximately $g(0)\sigma_0$, the inertia starts to rotate and the friction decreases rapidly due to the Stribeck effect. The spring contracts, and the spring torque decreases. The inertia slows down and the friction torque increases due to the Stribeck effect and the rotation stops. The phenomenon repeats itself. The parameters of the friction model are shown in Table 1. Unfortunately, the passivity inequality (6) is not satisfied with these data.

Simulation of a direct implementation of the LuGre model using the integration algorithm DASSL with a relative tolerance[1] Tol=10^{-4} leads to wrong results: The break away torque is too high and there are non-physical oscillations in the computed friction torque. This result is understandable, because the step-size control of variable step integrators treat variables as zero, when they are below a predefined absolute tolerance. If no other information is available, this tolerance is usually selected as a multiple of the relative tolerance. Since z is in the order of 10^{-5}, the step-size control on z is practically switched off most of the time.

As to be expected, the simulation result is improved when a scaling for state variable z is introduced: The oscillations of the friction torque disappear while the inertia is rotating. The friction torque of the model with and without scaling using integration algorithm DASSL and a relative tolerance of Tol=10^{-4} is shown in Figure 2.

In Figure 5, the scaled derivative of the bristle deflection, *dz/dt*, is present. As can be seen, very sharp changes of this variable appear when changing from the sliding to the stuck region and vice versa. An integrator has to detect this sharp change to compute a correct solution. In order to give the integrator a hint to this situation, a state event is triggered in the Modelica model, whenever *dz/dt* changes sign, by introducing the mentioned auxiliary Boolean equation. This technique improves the quality of the LuGre model simulation further, although still a considerable difference in the behavior of the friction model is present when simulating with different tolerances. In Figure 3 the friction torque is shown for relative tolerances Tol=10^{-4} and Tol=10^{-6}. The difference between these two simulations is caused by different break away torques.

The break away torque is related to the dwell-time and the rate of increase of the applied torque. The dwell-time is the time between sticking and break away. Since the LuGre model is a dynamic model, a varying break away torque can be expected. The simulated break away torque for a relative tolerance Tol=10^{-4} is τ_B≈1.5 Nm, for a tolerance Tol=10^{-6} it is τ_B≈1.48 Nm. This result is also achieved with integration algorithms with fixed step-size.



Figure 1: Simulation setup for stick-slip motion with the LuGre friction model.

σ_0	10^5	[Nm/rad]
σ_1	316.23	[Nms/rad]
τ_v	0.4	[Nms/rad]
τ_C	1	[Nm]
τ_S	1.5	[Nm]
ω_S	0.001	[rad/s]

Table 1: Parameter values of the LuGre friction model.

[1] In Dymosim the absolute and relative tolerance of the state vector are equal [4].

In Figure 4 the angular velocity, the rotation angle of the inertia and the rotation angle of the spring are shown. The scaled average bristle deflection z and the derivative of the bristle deflection are shown in Figure 5. The scaling factor is $z_n=10^5$. These results were obtained with the integration algorithm DASSL and a tolerance $Tol=10^{-6}$. They agree qualitatively with the results in [2].



Figure 2: Friction torque of the LuGre model without scaling (solid line) and with scaling (dashed line).



Figure 3: Friction torque of the LuGre model with DASSL and tolerance $Tol=10^{-4}$ (solid line) and $Tol=10^{-6}$ (dashed line).



Figure 4: Angular velocity (solid line), rotation angle (dashed line) of the inertia and rotation angle of the spring (dotted line).



Figure 5: Scaled average bristle deflection (top) and scaled derivative of average bristle deflection (bottom).

1.3 Comparison of LuGre and Modelica BearingFriction Model

In the Modelica standard library, friction element BearingFriction is present which constrains the relative angular acceleration $d\omega/dt$ to zero, when the relative speed ω vanishes, i.e., the relative movement is forced to stay at rest. This is in contrast to the more detailled LuGre model which describes also the small relative movements in the stuck region.

The model of Figure 1 was simulated by replacing the LuGre model with component BearingFriction from library Modelica.Mechanics.Rotational. The parameters of the LuGre model are shown in Table 1. The Stribeck velocity is rather small which corresponds to a sharp decrease in the friction torque when the inertia starts to rotate. Therefore a simplified parameterization for the component BearingFriction is used. The Coulomb friction is set to $\tau_C=1$ Nm and the stiction torque is set to $\tau_S=1.5$ Nm. Therefore the parameter setting in the BearingFriction model is: *tau_pos*=[0, 1; 1, 1.4] and *peak*=1.5.

To achieve accurate results for simulations with the LuGre model, a relative tolerance of $Tol=10^{-6}$ is needed. The number of output intervals was set to N=2500 for a simulation time of T_S=25 s. For all simulations this parameter setup was kept constant. The model was built up with Dymola version 4.1d, 2001-07-11. To get similar integration times for two simulations the executable Dymosim file was executed in the DOS window. The simulations were performed on a 2xPentium III with 600 MHz and 768 MB RAM.

The model was compiled with the Microsoft Visual C++ compiler (version 4.1). Compiling the model with the GNU compiler (version egcs 2.91.66) and the standard option setting in Dymola requires a very small integration step size of T_I=50 µs for the LuGre model when using the integration algorithm RKFIX4 (= Runge-Kutta method of order 4 with fixed step size), to obtain a converging result. A much larger step size is possible by the GNU compiler options -O3 (max. optimization) or -O0 (no optimization). The very small integration step size is needed for compiler options -O1 (Dymola default), -O2, and -Os. The reason for this strange behaviour could not be determined.

Using the Microsoft Visual C++ compiler with Dymola defaults, a converging result with the LuGre model and integrator RKFIX4 requires a maximum integrator step size of T_I=1 ms and for the explicit Euler method with fixed step size (EULER) T_I=0.9 ms. To achieve a converging result for mixed mode integration, a maximum integrator step size of T_I=2 ms is required. Mixed mode integration is a special Dymola technique for real-time simulation where fast states are discretized with the implizit and slow states with the explicit Euler method. State variable z from the LuGre model was used as “fastState”. Simulation statistics are summarized in Table 2.

For the model with BearingFriction, the maximum integrator step size is T_I=7 ms for RKFIX4 and T_I=4 ms for EULER, respectively. Variations in the maximum possible integration step size for various GNU compiler options was not observed. For the comparision, the same step sizes as in the LuGre model were used for the integrators with fixed step size. The simulation statistics are summarized in Table 3.

The difference in the CPU-time for integration of the LuGre model and the BearingFriction model is not significant due to the simple model. For more complex models, the CPU-time for the variable step-size integrator DASSL is related to the number of function and Jacobian evaluations which are about 5 to 6 times higher for the LuGre model as with the BearingFriction model. The differences in the number of Jacobian evaluations and in the minimum step sizes can be traced back to the stiffness of the differential equations of the LuGre model.

A comparison of the friction torque of the BearingFriction model and of the LuGre model is shown in Figure 6. The difference of these two simulated friction torques is a result of different break away torques of the models. For the LuGre model the break away torque is varying and it is approximately $\tau_B \approx 1.48$ Nm, for the BearingFriction model the break away torque is constantly τ_B=1.5 Nm. The peaks in the simulated friction torque when the inertia stops do not appear in the BearingFriction model because of the simplified parameterization and neglection of the Stribeck effect.

Integration algorithm	DASSL	RKFIX4	EULER	MIXED MODE
CPU-time for integration [s]	0.909	1.23	1.09	0.875
CPU-time for one GRID interval [ms]	0.363	0.494	0.436	0.360
No. of result points	2587	2537	2586	2531
No. of steps	1732	25000	30000	12500
No. of F-evaluations	5626	100000	30000	12500
No. of H-evaluations	4962	25020	30044	12518
No. of Jacobian-evaluations	595	-	-	-
No. of state events	43	19	43	17
Min. integration step-size [s]	$1.58 \cdot 10^{-7}$	10^{-3}	$0.1 \cdot 10^{-3}$	$2 \cdot 10^{-3}$
Max. integration step-size [s]	1.19	10^{-3}	$0.9 \cdot 10^{-3}$	$2 \cdot 10^{-3}$
Max. integration order	5	4	1	1

Table 2: Simulation statistics of the LuGre model.

Integration algorithm	DASSL	RKFIX4	EULER
CPU-time for integration [s]	0.847	1.24	1.22
CPU-time for one GRID interval [ms]	0.339	0.498	0.489
No. of result points	2519	2517	2519
No. of steps	303	25000	30000
No. of F-evaluations	905	100000	30000
No. of H-evaluations	2864	25010	30010
No. of Jacobian-evaluations	120	-	-
No. of state events	9	9	9
Min. integration step-size [s]	$1.79 \cdot 10^{-6}$	10^{-3}	$0.1 \cdot 10^{-3}$
Max. integration step-size [s]	4.1	10^{-3}	$0.9 \cdot 10^{-3}$
Max. integration order	5	4	1

Table 3: Simulation statistics of the BearingFriction model.



Figure 6: Friction torque of the LuGre model (solid line) and of the BearingFriction model (dashed line).

2 LuGre Clutch Models

2.1 Model Derivation

2.1.1 Clutch with LuGre Friction

Friction torque τ in clutches is usually described as a function of the friction coefficient $\mu(\omega_r)$, which is in turn a function of the relative angular velocity ω_r, of the normal force F_n, and of a geometry constant c_{geo} which takes into account the geometry of the device and the assumptions on the friction distributions:

$$\tau = \mu(\omega_r) c_{geo} F_n \tag{11}$$

The LuGre model [2] can be adapted to such a clutch description:

$$\frac{dz}{dt} = \omega_r - \frac{|\omega_r|}{g(\omega_r)} z \tag{12}$$

$$g(\omega_r) = \frac{1}{\sigma_0}\left(\mu_C + (\mu_S - \mu_C) e^{-(\omega_r/\omega_s)^2}\right) \tag{13}$$

$$\tau = \left(\sigma_0 z + \sigma_1 \frac{dz}{dt} + \sigma_2 \omega_r\right) c_{geo} F_n \tag{14}$$

where μ_C is the Coulomb friction coefficient and μ_S is the static friction coefficient. This model is related to the lumped dynamic tire model in [3].

In the Modelica model, the normal force F_n is provided as input signal u in a normalized form,

$$F_n = F_{n\,\max} \cdot u \tag{15}$$

where the maximum normal force $F_{n\,\max}$ is provided as parameter.

If the relative angular velocity does not vanish at simulation start, the initial value of z, see (12), should be computed such that $dz/dt = 0$ for $t = 0$ to avoid peaks in the friction torque at the start of the simulation. Again, a Boolean auxiliary equation is introduced to trigger an event at sharp changes of z, i.e., when dz/dt changes sign. As the average bristle deflection is small compared to other state variables, also a scaling is introduced for z.

2.1.2 One-Way-Clutch with LuGre Friction

A one-way-clutch is an element where a clutch is connected in parallel to a free wheel. This special element is needed to resolve the ambiguity of the friction torque which would be present if a free wheel would be explicitly connected in parallel to a clutch component. If the clutch is locked, the friction torque is computed by

$$\tau = c_{\max}\left(\sigma_0 z + \sigma_1 \frac{dz}{dt} + \sigma_2 \omega_r\right) F_{n\,\max} \tag{16}$$

where $c_{\max}$ has to be provided as parameter. With this parameter the maximum friction torque is defined, when the clutch is locked, a reasonable choice is

$$c_{\max} = 100\,.$$

The clutch is locked when the average bristle deflection z is negative. All other equations are similar to the clutch shown in the section above.

2.2 Comparison of Clutch Models

The behaviour of the LuGre models for clutches is compared to the clutch models of the Modelica standard library.

2.2.1 Simplified automatic Gearbox

A simplified model of a 3-gear automatic gearbox is simulated. The model with components from the standard Modelica rotational library is shown in Figure 7. The parameters of the clutches and brakes are shown in Table 4 and the gear shift table in Table 5. A switching sequence from first to third gear within T_S=4 s is simulated using N=4000 output intervals and integrator DASSL with a relative tolerance Tol=10^{-6}.

The same model with the LuGre clutch is shown in Figure 8. Brakes are replaced by series connection of a LuGre clutch with a fixed flange. The parameters of the LuGre clutches are shown in Table 6. The model with LuGre clutches is unstable when using integration algorithms with a fixed step-size, even if the step-size is very small.



Figure 7: 3-gear automatic gearbox model.

mue_pos	[0, 0.5]	-
peak	1	-
cgeo	1	-
fn_max	10	[N]

Table 4: Parameters of the standard clutches and brakes.

gear	C1	C2	B1	B2
0				
1	on		on	
2	on			on
3	on	on		

Table 5: Gear shift table of clutches and brakes.

σ_0	10^5	[m/rad]
σ_1	300	[ms/rad]
σ_2	0	[ms/rad]
μ_C	0.5	[Nm/N]
μ_S	0.5	[Nm/N]
c_{geo}	1	-
$F_{n\ max}$	10	[N]
ω_S	0.001	[rad/s]

Table 6: Parameters of the LuGre clutches.



Figure 8: 3-gear automatic gearbox model with LuGre clutches.

The two different implementations of the clutches show a similar behaviour. The angular velocity of the load with standard clutches is shown in Figure 9, in comparison to the model with the LuGre clutch the switching from one gear into the next is slightly slower. The friction torques of the standard and the LuGre clutches are shown in Figure 10. There is no difference in the friction torque for clutch C2. For the LuGre clutch C1 peaks appear at the switching points. The friction torques of the brakes and the LuGre clutches are shown in Figure 11. The friction torque of the LuGre clutch B1 shows peaks at the switching points and when the clutch gets stuck. The friction torque of the LuGre clutch B2 shows some small oscillations after the switching point. With (11) and the actual parameterization for the standard clutches the maximum friction torque of the clutch is τ_{max}=5 Nm. This is not always fulfilled with the LuGre clutches.



Figure 9: Angular velocity of load.



Figure 10: Friction torques of LuGre and standard clutches.



Figure 11: Friction torques of LuGre clutches and brakes.

The effect on the friction torque of a decreased stiffness of the bristles of σ_0=10^4 is shown in Figure 12 (the damping of the bristles is kept constant at σ_1=300). The amplitude of the peaks of the LuGre clutch C1 is bigger. The peak when clutch B1 gets stuck is smaller. The comparison of the angular velocity of the load with the standard model shows that the time for switching is decreased.

The effect on the friction torque of a decreased damping of the bristles σ_1=30 is shown in Figure 13 (the stiffness is kept constant σ_0=10^5). The height of the peaks of the LuGre clutch C1 is smaller when the clutch is activated. However, the amplitudes of the peaks when the clutches get stuck are increased and there are small oscillations. The comparison of the angular velocity of the load with the standard model shows that the time for switching is almost identical, but there are small oscillations in the angular velocity after the switching.

The angular acceleration of the load with standard clutches and brakes and with LuGre clutches with different dynamic parameters is shown in Figure 14. The maximum acceleration with LuGre clutches is higher than with standard clutches and brakes. Increasing the stiffness of the bristles in the LuGre clutches results in higher peaks at the switching points. When the damping of the bristles is reduced, oscillations occur and the acceleration is negative.



Figure 14: Acceleration of load with standard clutches and brakes (top left), LuGre clutches and dynamic parameters σ_0=10^5 and σ_1=300 (top right), dynamic parameters σ_0=10^4 and σ_1=300 (bottom left), and dynamic parameters σ_0=10^5 and σ_1=30 (bottom right).



Figure 12: Friction torque of LuGre clutches C1, C2, B1, and B2 with dynamic friction parameters σ_0=10^4 and σ_1=300.



Figure 13: Friction torque of LuGre clutches C1, C2, B1, and B2 with dynamic friction parameters σ_0=10^5 and σ_1=30.

2.2.2 Car Model with Automatic Gear

A complete model of a car power train with an automatic gearbox was simulated. The model of the automatic gearbox is shown in Figure 15, for details see [7]. All clutches are based on the LuGre friction model. The parameters of the LuGre clutches are shown in Table 7. The maximum normal force is not shown, because it is different for each clutch. For the simulation of T_S=200 s, integrator DASSL was used with a relative tolerance Tol=10^{-4} and N=1000 output intervals. The model with LuGre clutches is unstable when using integration algorithms with a fixed step-size, e.g. RKFIX4, even with very small step sizes. The model is also unstable with integration algorithm DASSL when the stiffness of the bristles is reduced to σ_0=10^4, and the damping of the bristles is in the range $0.003 \le \sigma_1 \le 3$. When the stiffness of the bristles is increased, the simulation time also increases, as to be expected.

The *velocity* of the car with standard clutches and with LuGre clutches is shown in Figure 16. There is hardly any difference in the velocity, except at the gear shift at $t\approx 122$ s, where the velocity is decreasing with the LuGre clutches (which is qualitatively wrong) in contrast to the model with standard clutches where the velocity is not decreasing. The *acceleration* of the car with standard clutches and with LuGre clutches is shown in Figure 17. There are small differences between the two curves, especially at $t\approx 122$ s where the velocity of the car is decreasing. The maximum acceleration with the LuGre clutches is $a_{max}\approx 10^5$ m/s^2, a completely unrealistic value. The amplitude of these peaks depends on the dynamic parameters. When the damping of the bristles is increased or the stiffness of the bristles is decreased the amplitude of the peaks increases.

σ_0	10^5	[m/rad]
σ_1	0.03	[ms/rad]
σ_2	0	[ms/rad]
μ_C	0.12	[Nm/N]
μ_S	0.144	[Nm/N]
c_{geo}	1	-
ω_S	0.5	[rad/s]

Table 7: Parameters of the LuGre clutches.



Figure 15: Automatic gearbox model.



Figure 16: Velocity of car with standard clutches (top) and LuGre clutches (bottom).



Figure 17: Acceleration of car with standard clutches (top) and LuGre clutches (bottom).

Conclusions

Modelica models for bearing friction, clutches and one-way clutches have been implemented based on the recently developed new LuGre friction model [2]. This model has been used successfully for controller design to compensate friction effects [6]. The friction model seems to be well suited for this purpose, because the (stiff) differential equation description allows an easier application of standard theory.

In this article it was investigated whether the LuGre friction model has also advantages when used in simulations. Especially, the potential for real-time simulation of the shift dynamics of automatic gearboxes was evaluated. It turns out that reasonable results can be achieved if one friction element is contained in a model. As to be expected, the simulation time is higher, since a stiff differential equation is solved. A potential advantage seemed to be that no events occur in the model because handling of state events is always problematic in real-time simulation. However, it seems to be that state events (or something similar) have to be artificially introduced to detect the sharp changes in z, in order to arrive at reliable simulations.

When friction elements are dynamically coupled, as it is the case in automatic gearbox models, LuGre based clutch models seem to be not suited: Fixed step size integrators, as needed for real-time simulation, could not be applied in the two test cases, because it was not possible to get stability (even for very small step sizes). Using the usually very robust and reliable variable step-size integrator DASSL, resulted in simulations which are quite sensitive on the choice of the dynamic LuGre friction parameters, and even leaded to instabilities in certain parameter ranges.

Bibliography

[1] N. Barabanov and R. Ortega, "Necessary and Sufficient Conditions for Passivity of the LuGre Friction Model", *IEEE Transactions on Automatic Control,* Vol. 45, 2000.

[2] C. Canudas de Wit, H. Olsson, K. J. Åström and P. Lischinski, "A New Model for Control of Systems with Friction", *IEEE Transactions on Automatic Control*, Vol. 40, 1995.

[3] C. Canudas de Wit, P. Tsiotras, "Dynamic Tire Friction Models for Vehicle Traction Control", *Conference on Decision and Control,* Phoenix, Arizona, Dec. 1999.

[4] Dymola, *Dynasim AB,* Lund, Sweden, version 4.1, http://www.dynasim.se.

[5] M. Gäfvert, "Comparison of two Friction Models", *Master thesis, Lund Institute of Technology,* University of Lund, 1996.

[6] H. Olsson, "Control of Systems with Friction", *Phd. thesis, Lund Institute of Technology,* University of Lund, 1996.

[7] M. Otter, C. Schlegel and H. Elmqvist, "Modeling and Realtime Simulation of an Automatic Gearbox using Modelica", *Proc. ESS'97 European Simulation Symposium,* Passau, Germany, pp. 115-121, 1997.

Appendix

Simplified rotational LuGre Model

```
model LuGreRot "LuGre bearing friction"
  import R=Modelica.Mechanics.Rotational;
  import SI = Modelica.SIunits;
  extends R.Interfaces.Rigid;

  parameter SI.Torque tau_s=1.5;
  parameter SI.Torque tau_c=1;
  parameter Real tau_v=0.4 "Nms/rad";
  parameter SI.AngularVelocity ws=0.001;
  parameter Real sigma0=1e5 "Nm/rad";
  parameter Real sigma1=316.22 "Nms/rad";
  parameter Real zN=1.e-5 "Nom. value";

  SI.AngularVelocity w "Abs. speed";
  SI.AngularAcceleration a "dw/dt";
  SI.Angle z(start=0) "Bristle defl.";
  SI.AngularVelocity zs(start=0) "dz/dt";
  SI.Torque tau "Friction torque";

  Real g "see [2]";
  SI.Angle zStart "Start value of z";
  Boolean zsneg "Trigger events";
equation
  // Initial conditions
  when initial() then
    zStart = sign(w)*g/zN;
    reinit(z, zStart);
  end when;

  // Speeds and accelerations
  w = der(phi);
  a = der(w);

  // Deflection of bristles
  zs = der(z);
  zs = (w/zN - abs(w)*z/g);

  // Friction torque
  g = 1/sigma0*(tau_c + (tau_s - tau_c)*
               exp(-(w/ws)^2));
  tau = sigma0*z*zN+sigma1*zs*zN+tau_v*w;

  // Equilibrium of torques
  flange_a.tau + flange_b.tau - tau = 0;

  // Trigger events when dz/dt=0
  zsneg = zs < 0;
end LuGreRot;
```

Clutch with LuGre Friction

```
model ClutchLuGre "LuGre Clutch model"
  import R=Modelica.Mechanics.Rotational;
  import SI = Modelica.SIunits;
  import B = Modelica.Blocks.Interfaces;
  extends R.Interfaces.Compliant;

  parameter Real mue_s=1.5;
  parameter Real mue_c=1;
  parameter SI.AngularVelocity ws=0.001;
  parameter Real sigma2=0.1;
  parameter Real sigma0=1e5;
  parameter Real sigma1=316.22;
  parameter Real sigma2=0.1;
  parameter Real zN=1.e-5 "Nom. value";
  parameter Real cgeo = 1 "Geom. const."
  parameter SI.Force fn_max = 1;

  B.InPort inPort(final n=1);

  SI.AngularVelocity w_rel "Rel. speed";
  SI.AngularAcceleration a_rel;
  SI.Angle z(start=0) "Bristle defl.";
  SI.AngularVelocity zs(start=0) "dz/dt";
  SI.Torque tau "Friction torque";
  SI.Force fn "Normal force(=fn_max*u)";

  Real g "see [2]";
  SI.Angle zStart "Start value of z";
  Boolean zsneg "Trigger events";
  Boolean free;
  Real u "normalized force input [0..1]";
equation
  // Initial conditions
  when initial() then
    zStart = sign(w_rel)*g/zN;
    reinit(z, zStart);
  end when;

  // Relative quantities
  w_rel = der(phi_rel);
  a_rel = der(w_rel);
```

```
  // Normal force and frict. for fn <= 0
  u = inPort.signal[1];
  free = u <= 0;
  fn = if free then 0 else fn_max*u;

  // Deflection of bristles
  zs = der(z);
  zs = if free then 0 else
         (w_rel/zN - abs(w_rel)*z/g);
  g = 1/sigma0*(mue_c + (mue_s - mue_c)*
                exp(-(w_rel/ws)^2));
  // Friction torque
  tau = if free then 0 else ceo*fn*
         (sigma0*z*zN + sigma1*zs*zN +
          sigma2*w_rel);

  // Trigger events when dz/dt=0
  zsneg = zs < 0;
end ClutchLuGre;
```

One-Way-Clutch with LuGre Friction

```
model OneWayClutchLuGre
            "Freewheel and clutch"
  import R=Modelica.Mechanics.Rotational;
  import SI = Modelica.SIunits;
  import B = Modelica.Blocks.Interfaces;
  extends R.Interfaces.Compliant;

  parameter Real mue_s=1.5;
  parameter Real mue_c=1;
  parameter SI.AngularVelocity ws=0.001;
  parameter Real sigma2=0.1;
  parameter Real sigma0=1e5;
  parameter Real sigma1=316.22;
  parameter Real sigma2=0.1;
  parameter Real zN=1.e-5 "Nom. value";
  parameter Real cgeo = 1 "Geo. const."
  parameter SI.Force fn_max = 1;

  B.InPort inPort(final n=1);

  SI.AngularVelocity w_rel "Rel. speed";
  SI.AngularAcceleration a_rel;
  SI.Angle z(start=0) "Bristle defl.";
  SI.AngularVelocity zs(start=0) "dz/dt";
  SI.Torque tau "Friction torque";
  SI.Force fn "Normal force(fn_max*u)";

  Real g "see [2]";
  SI.Angle zStart "Start value of z";
  Boolean zsneg "Trigger events";
  Boolean free;
  Boolean locked;
  Real u "normalized force input [0..1]";
protected
  constant Real cmax=100;
equation
  // Initial conditions
  when initial() then
    zStart = sign(w_rel)*g/zN;
    reinit(z, zStart);
  end when;

  // Relative quantities
  w_rel = der(phi_rel);
  a_rel = der(w_rel);

  // Normal force and frict. for fn <= 0
  u = inPort.signal[1];
  free = u <= 0;
  fn = if free then 0 else fn_max*u;
  locked = z < 0;

  // Deflection of bristles
  zs = der(z);
  zs = (w_rel/zN - abs(w_rel)*z/g);

  // Friction torque
  g = 1/sigma0*(mue_c + (mue_s - mue_c)*
               exp(-(w_rel/ws)^2));
  tau = if locked then
            fn_max*cmax*(sigma0*z*zN
         + sigma1*zs*zN + sigma2*w_rel)
       else (if free then 0 else
           cgeo*fn*
          (sigma0*z*zN + sigma1*zs*zN +
           sigma2*w_rel));
end OneWayClutchLuGre;
```